
Abstract 
Recently, I have published a paper, ‘How PROC SQL and SAS

®
 macro programming made my statistical analysis 

easy? A case study on linear regression’ (Refer CinSUG website for Ohio SAS Users Conference 2015 for a full 
paper). In that paper, various macro programs were created to eliminate outlier data during normalization of non-
parametric data. Often, this outlier data is valuable and provides a different outlook while drawing conclusions from 
the entire set of data after analysis. The objective of this paper is to show a technique of rescuing non-parametric 
outlier data using a non-parametric test after the analysis of parametric portion of the data using SAS. This paper also 
explains how conclusions are drawn from both, parametric and non-parametric tests without sacrificing the outlier 
data.  
  

1. Introduction 
 

Figure 1 

 
For many scientists and data analysts, outliers are like a ‘black box’ in conventional statistics. Many believe that these 
outlier observations arise due to errors or due to improper procedures in the experiment. Majority of them eliminate 
the outliers unscientifically by brute force. Some identify them statistically but discard them as if they are junk. Some 
understand importance of the outliers but they do not know how to deal with it. If you are one among them or 
interested in scope of the outliers, then this paper is the right resource for you. Outliers are like hidden treasure in 
data analytics. Discarding true outliers from data may costs huge amount of money in certain projects such as clinical 

trials. An innovation with true outliers in data analytics using SAS is shown in Figure 1. In this figure, a simplified 

hypothetical data on age vs height is shown for visualization. Two statistical outliers are clearly marked in the 
pictograph. At about 3 years of age, one outlier’s height is equal to 18-20 years old person. As opposed to this, at 
about 18 years of age, another outlier’s height is approximately equal to 3 years old kid. This dataset can be 
normalized by eliminating these two statistical outliers. As usual, conventional statistics can be performed on the 
normalized dataset. On the other hand, rescued statistical outliers can be tested with non-parametric tests such as 
Kruskal-Wallis one-way ANOVA. Stringent checks can be performed on outcome and true outliers can be identified. 
For example, in this hypothetical dataset, out of two, one outlier can be a dwarf. In medical field, this condition is 
called as dwarfism. In other words, an 18 years old person with height equal to a 3 years old kid is not an 
experimental or procedural error. This is a simple hypothetical example of an innovation with true outliers. On the 
other hand, it is possible that the 3 years old kid with height equal to 18 years appears to be an experimental or 
procedural error or an error during data processing.  
 
In this paper, a raw data on two interrelated plant metabolites (X and Y) is utilized for analysis. This data is obtained 
from SHEET1 of DATAXY.XLS, a Microsoft Excel 97-2003 file (See Table 1 and Excel SHEET1 snapshot in 

Appendix). This raw data is transformed, normalized and utilized for generating a linear regression model. While 
normalization, outlier data is separated and rescued for further non-parametric analysis using Kruskal-Wallis test. 
Finally, conclusions are drawn from both, parametric and non-parametric portions of the data. Significance and 
applications of rescuing outlier non-parametric data are discussed. Analysis was carried out using SAS

®
 9.4 software 

in Windows operating system. To reduce coding, various SAS macros created earlier were utilized in this paper 
(Perla, 2005). Refer following paper for background and SAS macros.   

 

 

 
 

2. Data Import 
Macro variable ‘PATH’ is created for Excel file path. Macro ‘EXCEL_IMPORT’ utilizes macro variable ‘PATH’ while 
importing Excel file ‘DATAXY’ into SAS. The output SAS dataset created after execution of macro ‘EXCEL_IMPORT’ 
is ‘XY’ (Table 2).  All the defined macros are stored under the folder ‘STATMACROS’. These macros are invoked by 

‘OPTIONS’ statement mentioned below. 
 

%let path= C:\Users\Documents\SAS\My SAS Files\; 

options mautosource sasautos="C:\Users\Documents\SAS\statmacros\";  

 

 

%excel_import (excel_file=dataxy, excel_sheet=sheet1, dataset=xy); 

**Refer Perla (2015) for macro code; 

http://www.cinsug.org/sites/g/files/g1233521/f/201506/Venu Perla How PROC SQL and SAS%C2%AB Macro Programming Made My Statistical Analysis Easy A Case Study on Linear Regression.pdf

Table 2 

 
 

3. Preliminary Data Analysis 
Execution of macro ‘SCATTER_CORR’ produces scatter plot and Pearson correlation coefficients for X- and Y-
variables of the dataset ‘XY’. Output indicates that the relationship between X and Y is weak (Pearson correlation 
coefficient: 0.15) (Figure 2; Table 3). 

 
%scatter_corr (dataset=xy, xvar=x, yvar=y); 

**Refer Perla (2015) for macro code; 

 

 

Figure 2 
 

Table 3 

 
 

Output produced after execution of the macro ‘REG_NORMALITY’ suggests that all the tests of normality are 
significant and the distribution of residuals for Y is not normal (Figure 3; Table 6). Although, the adjusted R

2
 value is 

negligible, ‘LACK OF FIT’ for linear model is not significant. This indicates that the relationship between X and Y can 

be explained by linear regression model (Table 4 and 5). However, data has to be normalized before developing a 

linear regression model for X and Y in conventional statistics. 
 
%reg_normality (dataset=xy, xvar=x, yvar=y); 

**Refer Perla (2015) for macro code; 

 

Table 4 

 
 

Table 5A 

 
 

Table 5B 

Figure 3 

 
Table 6 

 

 

4. Data Normalization 
 
4.1. Data Transformation  

Data transformation was carried out using Box-Cox power transformation technique (Perla, 2015). The best lambda, 
which is -0.25, is identified by invoking the macro ‘BOX_COX_LAMBDA’ on non-zero and non-negative dataset ‘XY’ 
(Table 7).  This lambda is the exponent to be used for transforming the Y-variable.  

 

%box_cox_lambda (pre_trans_dataset=xy, xvar=x, yvar=y); 

**Refer Perla (2015) for macro code; 

Table 7 

 
 

In biological sciences, it is advisable to adopt a transformation that is commonly used by others in the field. Since the 
best lambda value is close to ‘0’, logarithmic transformation is adopted for Y-variable in this paper.  Macro 
‘TRANSFORM_LAMBDA_2,’ a modified version of macro ‘TRANSFORM_LAMBDA’ (Perla, 2015), is explained in 
Appendix. Global macro variable ‘OTHERVARS’ is a comma separated list of variables to be included in the output 

dataset ‘XY_TRANS’. Variable ‘ZERO_Y‘ in the output dataset represents Y-variable after logarithmic transformation 
(Table 8). Variable ‘ZERO_Y’ is used for further analysis.  

 

%let othervars=id,x; 

%transform_lambda_2 (pre_trans_dataset=xy, yvar=y, trans_dataset= xy_trans); 

 
Table 8 

 
 
For a meaningful Y-intercept, X-variable is standardized using the macro ‘STDIZE_X’. The dataset ‘XY_TRANS’ 
obtained above is used here to get output dataset ‘XY1’. ‘XY1’ contains transformed Y-variable and standardized X-
variable (X) (Table 9).  
 
 %stdize_x (trans_dataset=xy_trans, trans_stdize_dataset=xy1, xvar=x); 

**Refer Perla (2015) for macro code; 

 

Table 9

4.2. Identification and Slicing of Statistical Outliers While Normalizing the Data 
Macro ‘REGRESSION_WOUT_OUTLIERS_2,’ a modified version of macro ‘REGRESSION_WOUT_OUTLIERS’ 
(Perla, 2015), is explained in Appendix. This master macro generates ANOVA, parametric estimates, distribution of 
residuals and tests of normality for the input dataset (DATASET or INDATA). Furthermore, output generated on 
outlier and influential observations is useful while identifying and slicing of outlier observations from the input dataset. 
There are four macros (REG_NORMALITY, OUTLIER_OBS, SLICE_OBS_2 and NO_OUTLIER_DATA) within this 
master macro. Except for macro ‘SLICE_OBS_2’, all the keyword parameters for other macros are described earlier 
by Perla (2015). Macro ‘SLICE_OBS_2’ is explained in Appendix. ‘R, INFLUENCE, RSTUDENTBYLEVERAGE, 

DFFITS, DFBETAS’ and ‘COOKSD’ options of the macro ‘OUTLIER_OBS’ generate detailed outlier and or influential 
observations for the input dataset. Outliers are sliced as per the criteria adopted by Perla (2015). 
 
Set of outlier observations to be sliced are referenced by a global macro variable ‘OBSET’. Master macro 
‘REGRESSION_WOUT_OUTLIERS_2’ is invoked twice, first without outliers (OBSET=0), then with the identified 
outliers from the first run. Outlier free dataset is generated after second run. This two-step process is continued until 
all the tests of normality and ‘LACK OF FIT’ options produce non-significant values. Perhaps, one should be careful 
while identifying and slicing statistical outliers in the dataset.   
 

%let obset=0; 

%regression_wout_outliers_2 (dataset=xy1, indata=xy1, sliced_data=sliced, 

outdata=xy2, xvar=x, yvar=zero_y); 

**From this run, it is clear that obs=9,10,11,12,169,170,171,172 are outliers; 

%let obset=9,10,11,12,169,170,171,172; 

**For improvement, slice obs=9,10,11,12,169,170,171,172 from XY1 dataset to get 

outlier free data; 

%regression_wout_outliers_2 (dataset=xy1, indata=xy1, sliced_data=sliced, 

outdata=xy2, xvar=x, yvar=zero_y); 

 

In the first cycle of execution shown above, results generated with ‘OBSET=0’ indicates that the transformed dataset 
is not normal (Figure 4; Table 10). However, model still holds good for X- and Y-variables (Table 11). As dataset is 
not normal, observation numbers for statistical outliers are identified from the results (Table 12; Figures 5-8), and the 

master macro is executed with ‘OBSET=9, 10, 11, 12, 169, 170, 171, 172’. Outlier-free output dataset (XY2) is tested 
again in the next cycle of execution of master macro. After several cycles of execution, normality tests and ‘LACK OF 
FIT’ values for dataset ‘XY5’ are non-significant (Table 13 and 14; Figure 9). Either dataset ‘XY5’ or  ‘XY6’ can be 

used for conventional statistics.    
 
 

 
Figure 4 

Table 10 

 
 

Table 11 

 
 

Table 12 

Figure 5 

 

 

 

 

 
Figure 6 

Figure 7 

 

 

 
Figure 8 

 

%let obset=0; 

%regression_wout_outliers_2 (dataset=xy2, indata=xy2, sliced_data=sliced1, 

outdata=xy3, xvar=x, yvar=zero_y); 

**From this run, it is clear that obs=21,22,23,24 are outliers; 

%let obset=21,22,23,24; 

**For improvement, slice obs=21,22,23,24 from XY2 dataset to get outlier free 

data; 

%regression_wout_outliers_2 (dataset=xy2, indata=xy2, sliced_data=sliced1, 

outdata=xy3, xvar=x, yvar=zero_y); 

%let obset=0; 

%regression_wout_outliers_2 (dataset=xy3, indata=xy3, sliced_data=sliced2, 

outdata=xy4, xvar=x, yvar=zero_y); 

**From this run, it is clear that obs=41,42,43,44 are outliers; 

%let obset=41,42,43,44; 

**For improvement, slice obs=41,42,43,44 from XY3 dataset to get outlier free 

data; 

%regression_wout_outliers_2 (dataset=xy3, indata=xy3, sliced_data=sliced2, 

outdata=xy4, xvar=x, yvar=zero_y); 

 

 

%let obset=0; 

%regression_wout_outliers_2 (dataset=xy4, indata=xy4, sliced_data=sliced3, 

outdata=xy5, xvar=x, yvar=zero_y); 

**From this run, it is clear that obs=153,154,155,156 are outliers; 

%let obset=153,154,155,156; 

**For improvement, slice obs=153,154,155,156 from XY4 dataset to get outlier 

free data; 

%regression_wout_outliers_2 (dataset=xy4, indata=xy4, sliced_data=sliced3, 

outdata=xy5, xvar=x, yvar=zero_y); 

 

 

%let obset=0; 

%regression_wout_outliers_2 (dataset=xy5, indata=xy5, sliced_data=sliced4, 

outdata=xy6, xvar=x, yvar=zero_y); 

 

Table 13 

Figure 9 

 
 

Table 14 

 
 

5. Linear Regression Model 
‘PARAMETER ESTIMATES’ obtained from dataset ‘XY5’ after execution of the macro 
‘REGRESSION_WOUT_OUTLIERS_2’ are used for modelling linear regression. Note that adjusted R

2
 value is 

increased from 0.02 to 0.73 after normalization of data by slicing 5 IDs (5 IDs X 4 Replications = 20 Observations). 
The relationship between X- and Y-variables in the parametric portion of data can be explained by linear regression 
model (2).    
 

1. Raw data (non-normal):    Y = 2.384 + 1.567X (Adjusted R
2
: 0.02) 

2. Normalized data (without 20 observations): Y = –0.080 + 0.496X (Adjusted R
2
: 0.73) 

 

6. Correlation Analysis of Parametric Portion of the Data 
Correlation analysis of normalized portion of the data is performed by calling again the macro ‘SCATTER_CORR’ for 
UNSTDIZED_X- and Y-variables of the dataset ‘XY6’. Note that Pearson correlation coefficient of raw data (Figure 2; 
Table 3) is increased from 0.15 to 0.88 after normalization (Figure 10; Table 15). 

 
%scatter_corr (dataset=xy6, xvar=unstdized_x, yvar=y); 

**Refer Perla (2015) for macro code; 

Figure 10 

 
Table 15 

 
 

7. Rescuing and Analyzing Statistical Outliers 
There are 204 observations in the raw data set. Out of which 20 are outliers. From the above normalization process, 
it is clear that approximately 10% of the observations are outliers in the raw data. This outlier data is generally 
ignored when predictions are made for whole data. In depth analysis of outliers may provide meaningful insights on 
the data. Often, all the outliers are not true outliers. For this reason, the sliced 20 outliers (5 IDs) are combined, 
sorted and analyzed using non-parametric one-way ANOVA (Kruskall-Wallis test) after incorporating one maximum 
and one minimum ID from parametric portion of the data for reference. Details of this analysis are given below 
separately for X- and Y-variable. 
 

7.1. Combining All the Outlier Datasets 
Outlier datasets (SLICED-SLICED3) are combined and sorted to get ‘SLICED_ALL’ dataset (Table 16). There are 20 

observations (5 IDs) in the ‘SLICED_ALL’ dataset. 
 

title "Combined dataset for all the outliers from SLICED-SLICED4 datasets"; 

data sliced_all; 

 set sliced sliced1 sliced2 sliced3; 

run; 

 

Proc sort data=sliced_all; 

 by id; 

run; 

 

proc print data=sliced_all; 

run; 

 



Venu Perla, Ph.D.                                             Philadelphia Area SAS Users Group (PhilaSUG) Fall 2015 Meeting; October 29, 2015  
                                                                      Penn State Great Valley School of Graduate Professional Studies, Malvern, PA, USA 

14 
 

Table 16 

 
 

7.2. X-Variable 
 

7.2.1. Identification of Maximum and Minimum IDs for X-Variable in the Parametric Dataset 
Maximum and minimum IDs for UNSTDIZED_X-variable in the parametric dataset ‘XY6’ are identified by using 
‘MEANS’ and ‘SORT’ procedures of SAS.  From the sorted output dataset ‘XY6_MAXMIN_X’, it is very much evident 
that ID 39 and ID 5 are the maximum and minimum IDs for UNSTDIZED_X-variable, respectively (Table 17).  
 

title "Identification of IDs with max and min X-mean values in parametric 

dataset"; 

proc means data=xy6 mean; 

class id; 

var unstdized_x; 

output out=xy6_maxmin_x mean=x_mean; 

run; 

 

Proc sort data=xy6_maxmin_x; 

 by descending x_mean; 

run; 

 

proc print data=xy6_maxmin_x; 

run; 

Table 17 

 
 

7.2.2. Dataset for Two IDs (ID 39 and ID 5) 
‘PROC SQL’ is used to create a table ‘X_MAX_MIN’ for ID 39 and ID 5 (Table 18). 

 
title "Dataset for two IDs with max and min X-values";  

proc sql; 

create table x_max_min as 

 select*  

 from xy6 

 where id in (39,5);    

quit; 

 

proc print data=x_max_min; 

run; 

 
Table 18

 
 
7.2.3. ‘X_MAX_MIN’ Dataset is Combined with ‘SLICED_ALL’ Outlier Dataset  
Now, two datasets, ‘X_MAX_MIN’ and ‘SLICED_ALL’, are merged to get ‘SLICED_ALL_2_X_MAX_MIN’ dataset. 
Maximum and minimum X-values of parametric data serve as reference points for statistical outliers in the combined 
dataset (Table 19). Dataset ‘SLICED_ALL_2_X_MAX_MIN’ is sorted to maintain the ID numbers in order (Table 20). 

   
title "Now add max and min X IDs to dataset SLICED_ALL"; 

data sliced_all_2_x_max_min; 

 set sliced_all 

     x_max_min; 

run; 

proc print data=sliced_all_2_x_max_min; 

run; 

 
 
 
 
 



Venu Perla, Ph.D.                                             Philadelphia Area SAS Users Group (PhilaSUG) Fall 2015 Meeting; October 29, 2015  
                                                                      Penn State Great Valley School of Graduate Professional Studies, Malvern, PA, USA 

16 
 

Table 19

 
 

title "To maintain ID order, sort the dataset"; 

proc sort data=sliced_all_2_x_max_min; 

 by id; 

run; 

proc print data=sliced_all_2_x_max_min; 

run; 

 
Table 20 

 
 
7.2.4. Kruskal-Wallis Test for Non-Parametric Portion of X-variable 
The Kruskal-Wallis one-way ANOVA is performed on ‘SLICED_ALL_2_X_MAX_MIN’ dataset using ‘PROC 
NPAR1WAY’ with ‘WILCOXON’ option. Distribution of Wilcoxon scores for ‘UNSTDIZED_X’ is presented in Figure 
11. There are 7 IDs in the Figure 11. Out of which, ID 39 and ID 5 are upper and lower limits from parametric data, 

respectively. Although, there are 5 outliers (ID 3, ID 7, ID 13, ID 42 and ID 43), only 2 outliers (ID 42 and ID 43) 
appear to be deviated from normal population of X-variable. Remaining 3 (ID 3, ID 7 and ID 13) are within the range 
of upper and lower limits of parametric data. In summary, present technique identified only 2 true potential outliers (ID 
42 and ID 43) for X-variable. 

 
title "Kruskal-Wallis Test for non-parametric portion of X-variable"; 

title "Two IDs (39 and 5) with maximum and minimum X-values from parametric 

portion of the data serve as reference for outliers"; 

proc npar1way data=sliced_all_2_x_max_min wilcoxon; **non-parametric-one-way; 

 class id; 

 var unstdized_x; 

run; 

Figure 11 

 
7.3. Y-Variable 
Similar to X-variable, analysis for Y-variable is carried out below for statistical outliers. 
 
7.3.1. Identification of Maximum and Minimum IDs for Y-Variable in the Parametric Dataset 
The maximum and minimum IDs for Y-variable in the parametric dataset ‘XY6’ are identified by using ‘MEANS’ and 
‘SORT’ procedures of SAS.  From sorted output dataset ‘XY6_MAXMIN_Y’, it is very much evident that ID 22 and ID 
11 are the maximum and minimum IDs for Y-variable, respectively (Table 21).  

 
title "Identification of IDs with max and min Y-mean values in parametric 

dataset"; 

proc means data=xy6 mean; 

class id; 

var y; 

output out=xy6_maxmin_y mean=y_mean; 

run; 

 

Proc sort data=xy6_maxmin_y; 

 by descending y_mean; 

run; 

 

proc print data=xy6_maxmin_y; 

run; 

 
Table 21 

7.3.2. Dataset for Two IDs (ID 22 and ID 11) 
‘PROC SQL’ is used to create a table ‘Y_MAX_MIN’ for ID 22 and ID 11 (Table 22). 

 
title "Dataset for two IDs with max and min Y-values";  

proc sql; 

create table y_max_min as 

 select*  

 from xy6 

 where id in (22,11);    

quit; 

 

proc print data=y_max_min; 

run; 

 
Table 22 

 
 

7.3.3. ‘Y_MAX_MIN’ Dataset is Combined with ‘SLICED_ALL’ Outlier Dataset  
Now, the two datasets, ‘Y_MAX_MIN’ and ‘SLICED_ALL’ are combined to get ‘SLICED_ALL_2_Y_MAX_MIN’ 
dataset. Maximum and minimum Y-values from parametric data serve as reference points for outliers in the combined 
dataset (Table 23). Dataset ‘SLICED_ALL_2_Y_MAX_MIN’ is sorted to maintain the ID numbers in order (Table 24). 
 

title "Now add max and min Y IDs to dataset SLICED_ALL"; 

data sliced_all_2_y_max_min; 

 set sliced_all 

  y_max_min; 

run; 

proc print data=sliced_all_2_y_max_min; 

run; 

 
Table 23 

 
 

title "To maintain ID order, sort the dataset"; 

proc sort data=sliced_all_2_y_max_min; 

 by id; 

run; 

proc print data=sliced_all_2_y_max_min; 

run; 

Table 24 

 
 

7.3.4. Kruskal-Wallis Test for Non-Parametric Portion of Y-variable 
The Kruskal-Wallis one-way ANOVA is performed on ‘SLICED_ALL_2_Y_MAX_MIN’ dataset using ‘PROC 
NPAR1WAY’ with ‘WILCOXON’ option. Distribution of Wilcoxon scores for Y is presented in Figure 12. There are 7 
IDs in the Figure 12. Out of which, ID 22 and ID 11 are upper and lower limits from parametric data, respectively. 

Although, there are 5 outliers (ID 3, ID 7, ID 13, ID 42 and ID 43), only 2 outliers (ID 3 and ID 42) appear to be 
deviated from the normal population of Y-variable. Remaining 3 (ID 7, ID13 and ID 43) are within the range of upper 
and lower limits of parametric data. In summary, present technique identified only 2 potential true outliers (ID 3 and ID 
42) for Y-variable. 
 

title “Kruskal-Wallis Test for non-parametric portion of Y-variable”; 

title “Two IDs (22 and 11) with maximum and minimum Y-values from parametric 

portion of the data serve as reference for outliers”; 

proc npar1way data=sliced_all_2_y_max_min wilcoxon; **non-parametric-one-way; 

 class id; 

 var y; 

run; 

 

 
Figure 12 

 

8. Outliers to be Scrutinized 
Above outlier analysis for X-variable indicates that ID 42 and ID 43 are two potential outliers to be scrutinized for their 
X-values in the data. Similarly, outlier analysis for Y-variable indicates that ID 3 and ID 42 are two potential outliers to 
be scrutinized for their Y-values in the data. Overall, ID 42 appears to be undisputed outlier for both, X- and Y-
variables.  Following are the partial list of checks to be performed on potential true outliers before confirming them as 
true outliers: 
 
 Errors while entering data. 

 Errors while processing the data. 
 Errors in wet lab analysis. 
 Errors while processing samples for wet lab analysis, and 
 Errors in sample collection. 
 
These checks may vary from project to project. If these statistical outliers are not aroused due to errors, then there 
might be significant reasons for these non-normal values.    
 

9. Applications  
In life sciences, true outliers may serve as a source for identifying unique mechanism, pathway, genotype, strain or 
variety. Similarly, these true outliers may play a role in innovation while analyzing data pertaining to science, clinical 
research, technology, internet, banking, finance, marketing and other similar sectors.  
 

10. Conclusions 
Summary of various SAS programming steps involved in rescuing statistical outliers are presented in Figure 13. In 

this paper, a data on two interrelated plant metabolites was utilized for analysis. Various macros that were previously 
defined were utilized with or without modifications while importing and normalizing data, and developing a linear 
regression model for the two plant metabolites. While normalizing the data, statistical outliers were scientifically 
eliminated. Then, a dataset for these non-parametric outliers was created. This dataset was analyzed separately for 
each variable using Kruskall-Wallis one-way ANOVA test. Prior to Kruskall-Wallis test, for each variable, maximum 
and minimum IDs from normalized portion of the dataset were incorporated with the outlier dataset for reference. This 
rescue technique helped in identifying Wilcoxon ranks for the statistical outliers. Furthermore, true outliers in each 
variable can be identified with proper stringent error checks. True outliers are the potential source of innovation not 
only in life sciences but also in other branches of science, technology and business. 
  

 
Figure 13 

1 0.16 0.16 

1 0.1444 0.1681 

1 0.1681 0.1681 

1 0.1521 0.1521 

2 0.4096 0.2209 

2 0.3364 0.2704 

2 0.3721 0.2401 

3 4.84 234.09 

3 4.41 222.01 

3 4 228.01 

3 5.29 225 

4 0.16 0.49 

4 0.1764 0.4225 

4 0.1521 0.5184 

4 0.1369 0.5041 

5 0.01 0.25 

5 0.0225 0.2025 

5 0.0121 0.2704 

5 0.0196 0.2916 

6 0.49 0.36 

6 0.4096 0.3025 

6 0.5329 0.4489 

6 0.4624 0.3721 

7 6.25 1.21 

7 5.76 1.4161 

7 7.84 0.9409 

7 6.76 0.8836 

8 0.16 0.25 

8 0.16 0.2116 

8 0.1764 0.2704 

8 0.1369 0.2401 

9 0.25 0.36 

9 0.3025 0.3969 

9 0.3844 0.3481 

9 0.2304 0.3844 

10 1.69 0.81 

10 1.44 0.7056 

10 1.8496 0.9025 

10 1.6641 0.7744 

11 0.16 0.16 

11 0.1225 0.16 

11 0.1936 0.1764 

11 0.1764 0.1296 

12 3.24 2.56 

12 2.89 2.89 

12 3.0276 2.6896 

12 3.3489 2.4964 

13 0.25 3.24 

13 0.25 2.9241 

13 0.25 3.3489 

13 0.1936 3.1329 

14 0.25 0.25 

14 0.36 0.2209 

14 0.2916 0.2916 

14 0.2209 0.3136 

15 0.49 0.49 

15 0.4096 0.4356 

15 0.5476 0.5476 

15 0.5625 0.4624 

16 0.09 0.49 

16 0.0576 0.4356 

16 0.1156 0.5184 

16 0.1024 0.5476 

17 1.96 0.81 

17 1.8769 0.7396 

17 2.0164 0.8649 

17 1.9881 0.8836 



Venu Perla, Ph.D.                                             Philadelphia Area SAS Users Group (PhilaSUG) Fall 2015 Meeting; October 29, 2015  
                                                                      Penn State Great Valley School of Graduate Professional Studies, Malvern, PA, USA 

23 
 

18 0.64 0.36 

18 0.5929 0.49 

18 0.6889 0.3136 

18 0.6561 0.3025 

19 1.69 1 

19 1.5129 0.9409 

19 1.7424 1.21 

19 1.8496 0.9025 

20 0.36 0.36 

20 0.3249 0.49 

20 0.36 0.36 

20 0.36 0.3249 

21 1.44 1 

21 1.44 1 

21 1.44 0.8836 

21 1 0.9216 

22 4 4.41 

22 4 5.76 

22 4.84 4 

22 3.24 3.8025 

23 0.49 0.36 

23 0.49 0.3136 

23 0.5476 0.3969 

23 0.4356 0.4225 

24 1.69 1.21 

24 1.69 1.21 

24 1 0.9216 

24 1.96 1.44 

25 1.21 1 

25 1.21 1 

25 1 0.9025 

25 0.8836 1.44 

26 4 1.69 

26 5.29 1.44 

26 4 1.69 

26 3.8416 1.7956 

27 0.36 0.49 

27 0.36 0.4096 

27 0.2916 0.5329 

27 0.4096 0.5476 

28 4.41 2.89 

28 4 2.7225 

28 4.84 2.9929 

28 5.29 2.9584 

29 3.24 1.96 

29 3.4596 2.1609 

29 3.2761 1.9044 

29 3.2041 1.9881 

30 1.44 0.64 

30 1.5376 0.7056 

30 1.44 0.5929 

30 1.3689 0.6561 

31 1 0.49 

31 0.9216 0.5929 

31 1.21 0.4624 

31 1.2544 0.5329 

32 4.41 2.25 

32 4.41 2.56 

32 4.41 2.56 

32 4 2.2201 

33 1.96 1 

33 1.8225 1.21 

33 2.0164 1.44 

33 2.0449 0.9025 

34 0.49 0.64 

34 0.4096 0.7056 

34 0.5184 0.64 

34 0.4624 0.6084 

35 0.25 0.25 

35 0.3364 0.3025 

35 0.2401 0.2601 

35 0.2601 0.2304 

36 0.81 0.49 

36 0.8836 0.5184 

36 0.7744 0.49 

36 0.7396 0.4624 

37 1.44 0.25 

37 1.44 0.3025 

37 1.69 0.2401 

37 1.21 0.2304 

38 1.21 0.49 

38 1.21 0.5476 

38 1 0.5329 

38 1.69 0.4624 

39 6.25 4 

39 6.0025 4.41 

39 6.4516 3.8809 

39 6.3001 4 

40 1 0.49 

40 1 0.49 

40 0.9216 0.5476 

40 1.21 0.4624 

41 0.81 0.64 

41 0.8649 0.6724 

41 0.8281 0.7056 

41 0.7744 0.5776 

42 9 7.29 

42 9 6.25 

42 8.2944 7.8961 

42 9.61 7.6176 

43 17.64 2.25 

43 18.49 2.2801 

43 17.0569 2.1609 

43 18.8356 2.4025 

44 0.81 1 

44 0.7569 1 

44 0.81 1 

44 0.81 1 

45 3.61 2.56 

45 3.8809 2.89 

45 3.5344 2.4649 

45 3.6481 2.6244 

46 1 0.64 

46 1 0.6724 

46 1 0.6084 

46 0.9216 0.5776 

47 1.44 0.49 

47 1.44 0.5329 

47 1.2321 0.4489 

47 1.7161 0.5329 

48 0.64 0.49 

48 0.64 0.5476 

48 0.6724 0.4356 

48 0.6084 0.4761 

49 1.96 0.64 

49 2.0449 0.6561 

49 1.8769 0.5776 

49 2.0449 0.6889 

50 1.96 1.96 

50 1.96 1.96 

50 1.9881 1.9044 

50 1.9044 2.0164 

51 1 1 

51 1.2321 1.0201 

51 0.9409 0.9216 

51 1.0201 0.9801 
 
 

 
Excel SHEET1 Snapshot. 

 
 

Macro ‘TRANSFORM_LAMBDA_2’: 
 
/************************************************************************************* 

transform_lambda_2: A SAS macro for transforming Y-values using lambda value. 

All possible power transformations are performed here using lambda value.  

 

Author: Venu Perla, Ph.D. 

Date: July, 2015. 

%let othervars = Comma separated list of variables to be included in 

trans_dataset. This global variable should be created before 

calling the macro '%TRANSFORM_LAMBDA_2'. 

pre_trans_dataset = Name of the dataset with non-zero and non-negative x- and y-

values. 

Xvar = Name of the X-variable. 

Yvar = Name of the Y-variable. 

trans_dataset = Name of the dataset for storing transformed data. 

*************************************************************************************/ 

%let othervars=           ; **do not forget comma between the variable names; 

%macro transform_lambda_2 (pre_trans_dataset= ,yvar= ,trans_dataset= ); 

 title "Transformation of &yvar.-variable with convenient lambda"; 

 proc sql; 

  create table &trans_dataset as 

  select  &othervars, &yvar,  

    1/(&yvar**2) as neg_2_&yvar, 

    1/(&yvar**1) as neg_1_&yvar, 

    1/(sqrt(&yvar)) as neg_half_&yvar, 

    log(&yvar) as zero_&yvar, 

    sqrt(&yvar) as half_&yvar, 

    &yvar**1 as one_&yvar, 

    &yvar**2 as two_&yvar 

  from &pre_trans_dataset; 

 quit; 

 

 proc print data=&trans_dataset; 

 run; 

%mend transform_lambda_2; 

 

*%let othervars=      ; 

*%transform_lambda_2 (pre_trans_dataset= ,yvar= ,trans_dataset= ); 

/************************************************************************************/ 

 

 
Macro ‘REGRESSION_WOUT_OUTLIERS_2’: 
 

/************************************************************************************ 

regression_wout_outliers_2: A master macro for identification and elimination 

outliers, and regression analysis of outlier-free data. 

 

Author: Venu Perla, Ph.D. 

Date: July, 2015  

 

Note: There are 4 macros in this macro. macro 'SLICE_OBS_2' is specific to this master 

macro.  

 

Dataset = Name of the dataset to be used regression analysis. 

Indata = It is same as DATASET.  

sliced_data = Name of the dataset for storing outlier observations. 

outdata = Name of the dataset for storing data after removing outlier 

observations. 

Xvar = Name of the x-variable to be used for analysis.  

Yvar = Name of the y-variable to be used for analysis.   

 

%let obset = ; Use this global macro variable to create a set of outlier 

observations before running the macro. 

&obset = Contains a set of observation numbers (not IDs) to be deleted from 

INDATA. Observations are separated by comma. 

****************************************************************************/ 

%macro regression_wout_outliers_2 (dataset= , indata= , sliced_data= , outdata= ,xvar= 

, yvar=); 

 %reg_normality (dataset=&dataset, xvar=&xvar, yvar=&yvar); 

 %outlier_obs (indata=&indata, xvar=&xvar, yvar=&yvar); 

 %slice_obs_2 (indata=&indata, sliced_data=&sliced_data); 

 %no_outlier_data (indata=&indata, sliced_data=&sliced_data, outdata=&outdata);  

%mend regression_wout_outliers_2; 

/***************************************************************************/ 

 
 

Macro ‘SLICE_OBS_2’: 
 
/************************************************************************************* 

slice_obs_2: A SAS Macro for deleting outlier/influencing observations from the 

dataset.  

 

Author: Venu Perla, Ph.D. 

Date: July, 2015. 

 

It utilizes OBSET= , an external global macro variable. One can assign more than one 

observations for macro variable 'OBSET='. 

 

indata = Name of the dataset to be used for slicing observations. 

sliced_data = Name of the output dataset for storing only the outlier observations.  

   

%let obset = Set of outlier observation number(s) separated by comma (observations to 

be removed from INDATA). 

 

Note: obset= is set to zero initially. When there are no observations to be sliced, 

then it will not produce any sliced observations. This will not affect further data 

processing in other macros. 

*************************************************************************************/ 

 

%macro slice_obs_2 (indata= ,sliced_data= ); 

 title "Dataset for outlier observation(s): &sliced_data"; 

 data &sliced_data; 

  do slice=&obset;  

   set &indata point=slice; 

   output; 

  end; 

  stop; 

 run; 

 

 proc print data=&sliced_data; 

 run; 

%mend slice_obs_2; 

/***********************************************************************************/ 

 


